
1 What is qfis

qfis (also QFIS) is a static program verifier for imperative programs working on
sequences of integers and annotated with formulas in the quantifier-free fragment
of the theory of integer sequences. qfis generates verification conditions from
an annotated program by backward reasoning. The verification conditions are
translated into SMT input files and CVC3 tries to discharge them. qfis reports
the results of verification back to users.

If you are familiar with program verifiers such as Boogie [4, 5] or Why [1, 3],
using qfis is a very similar experience, except that qfis does not work on generic
programs and first-order annotations but is targeted to the theory of quantifier-
free integer sequences described in [2]. You can try out qfis online without
installing any software through COMCOM at:

http://cloudstudio.ethz.ch/comcom/#QFIS

2 Downloading and installing

qfis is implemented in Eiffel and available for download in source and pre-
compiled form from:

http://bitbucket.org/caf/qfis/

Since version 1.1, qfis works on both GNU/Linux and Windows environments.
It is distributed under GNU GPL v. 3 or later versions.

After you have downloaded it, you can install qfis following these simple
instructions (which use Unix shell syntax and commands by default).

1. Move the downloaded package (e.g., qfis.tgz) to a directory of your
choice (e.g., /home/user/tools/):

mv qfis.tgz /home/user/tools/

∗The first release was v. 0.1, on 24 March 2011.

2. Move to the installation directory and unpack the package:

cd /home/user/tools ; gzip -cd qfis.tgz | tar xvf -

Unpacking will create a directory (e.g., /home/user/tools/qfis/) with
examples, prelude files, and a binary for your architecture.

3. If you want to compile qfis (you can get the sources from the same repos-
itory), you have to install the EiffelStudio compiler. See the instructions
on:

http://www.eiffel.com

qfis has been developed and compiled with EiffelStudio 7.0. To compile
the binary, launch the script

./build-linux.sh or build-windows.bat

according to your platform, in the installation directory. You may have
to edit qfis.ecf if the compiler can’t locate the libraries needed.

4. Add the path to the bin directory to the environment variable $PATH

(%PATH% in Windows environments):

export PATH=$PATH:/home/user/tools/qfis/bin/

5. Set the environment variable $QFIS_PRELUDE (%QFIS_PRELUDE% in Win-
dows environment) to the prelude directory:

export QFIS_PRELUDE=/home/user/tools/qfis/prelude/

Notice that the path stored by the environment variable must use forward
slashes even in Windows environments. If you don’t set the variable, qfis
will look for background theories in its working directory.

6. Install CVC3 and make sure it is reachable from the program path. This
version of qfis was tested with CVC3 v. 2.2. To install CVC3 under a
Debian-based distribution you can type:

sudo apt-get install cvc3

If you use Emacs, you may want to install the package cvc3-el as well.

7. To have syntax highlighting with any text editor supporting GTK source
view (such as gedit), copy the file qfis.lang in the directory grammar to
the appropriate directory. For Ubuntu and other similar distributions:

sudo cp qfis.lang /usr/share/gtksourceview-2.0/language-specs/

3 Input language

qfis’s input language includes both a simple modular imperative language, to
describe computations on sequences of integer variables, and an annotation lan-
guage, very close to the quantifier-free theory of integer sequences [2].

In the following BNF grammars, the meta-characters are “::=” (rule defini-
tion), “�” (alternative), “〈· · · 〉” (optional part); square and round brackets and
all other special characters are terminals; + denotes one or more repetitions.

3.1 Differences with the theory of sequences

There are some differences between the theory of integer sequences as defined
in [2] and the formulas allowed in qfis. The major differences are the following:

• The following table summarizes the way operators on sequences are en-
coded in ASCII in qfis. This manual uses, whenever possible, pretty print-
ing, but input files must use the ASCII encoding.

Description Native ASCII Pretty-printed
Concatenation ◦ ++ ◦

Sequence equality
.
= ==

.
=

Sequence inequality 6 .= =/= 6 .=
Implication ⇒ ==> =⇒

Iff ⇔ <==> ⇐⇒
Element selection S[3] S[3] S[3]

Subrange S[3, 0] S<<3:0>> S�3:0�
True, False >,⊥ True, False True, False

Ranges with mixed endpoints (negative and positive) or empty interval
(e.g. S[2:1]) are currently not supported. In a range or element selection,
1 denotes the first element, 2 the second; 0 denotes the last element, −1
the second-to-last, etc.

• Regular expressions are currently disallowed in qfis.

• Arithmetic is not applicable directly to sequences in qfis, but only to
singletons (which are handled as integers). In practice, this means that
every sequence appearing in an arithmetic operation or relation must be
projected (e.g., X[1] + 2 <Y[1] instead of X + 2 <Y). In this respect,
notice that the empty sequence is not treated as zero, and not accepted
at all in arithmetic expressions.

• qfis annotations allow for full integer arithmetic (including multiplication
and integer division) and comparison and other operations between two
integer variables. For example, you can specify that two sequences have
the same length.

• qfis axioms also allow for quantifiers.

The extensions beyond the quantifier-free fragment entail that the resulting
verification conditions are in general undecidable, unlike the original theory
[2]. Overall, this seemed a better trade-off for users, given that the underlying
SMT solving technology includes, on the one hand, several optimization that
introduce incompleteness, but works reasonably well, on the other hand, to

prove the validity of formulas (as opposed to finding out for sure that they are
invalid).

3.2 Variable naming

qfis is case sensitive. Every identifier starts with a letter, followed by letters,
digits, or the special characters _ and #. In addition, variables must be named
according to the following convention:

Integer variables: identifier starts with lowercase letter

Sequence variables: identifier starts with uppercase letter

Boolean variables: identifier ends with the extra character ?

Labels (of assertions): identifier starts with the extra character _ and ends
with the extra character :

Variables are declared with the following syntax:

VariableDeclaration ::= Declaration � Declaration ; VariableDeclaration
Declaration ::= IntegerIdentifierList : INTEGER

� SequenceIdentifierList : SEQUENCE
� BooleanIdentifierList : BOOLEAN

IntegerIdentifierList, SequenceIdentifierList, and BooleanIdentifierList are
comma-separated non-empty lists of integer, sequence, and Boolean identifiers
respectively.

3.3 Declarations and background theories

Annotations can mention logic functions. Every logic function must be declared;
the signature lists the function argument’s types, according to the syntax:

FunctionDeclaration ::= Identifier 〈(TypeList)〉 : Type 〈in ‘‘path’’〉
TypeList ::= Type � Type , TypeList

Type ::= INTEGER � SEQUENCE � BOOLEAN

The variable naming convention for variables extends to functions: a function
returning an integer (respectively, sequence, Boolean) must have a name suitable
for an integer (respectively, sequence, Boolean) variable.

A function with only a declaration is used as uninterpreted. If the in clause is
supplied, the CVC3 file path is included as theory of the function declared. qfis
performs no check that the theory in path (if any) is consistent with the func-
tion declaration. There are two other ways to introduce axioms for a declared
function.

Declarations directory. You can set a declarations directory, where qfis
searches for theories for declared functions. The declarations directory can
be set either by setting the environment variable QFIS_PRELUDE or with the
-d=path flag when calling qfis. See the installation instructions above for rec-
ommend settings for the declarations directory; if you don’t set it as suggested,
concatenation and length are also used as uninterpreted.

For each function declared function fun (T1, T2, ..., Tn): (T), qfis will look
for a file named declaration_fun_t_t1_t2_..._tn_.cvc in the declarations
directory, where each ti is 0 if Ti is SEQUENCE, 1 if Ti is BOOLEAN, and 2
if Ti is INTEGER.

Axiom declarations. You can introduce axioms that are translated into
CVC3. qfis allows quantifiers in axioms but not in annotation formulas. The
syntax for axioms is then:

AxiomDeclaration ::= axiom QuantifiedFormula
QuantifiedFormula ::= Formula

� Quantifier (VariableDeclaration) ( QuantifiedFormula )
Quantifier ::= forall � exist

where quantifier-free formulas are defined as:

Formula ::= AtomicFormula � ( Formula )
� not Formula � Formula Connective Formula

Connective ::= and � or � =⇒ � ⇐⇒
AtomicFormula ::= True � False � BooleanQuery
BooleanQuery ::= BooleanIdentifier � old BooleanIdentifier

� BooleanIdentifier ( IdentifierList )
� Sequence SequenceComparison Sequence
� Integer ArithmeticComparison Integer

SequenceComparison ::=
.
= � 6 .=

ArithmeticComparison ::= = � 6= � < �≤ � > � ≥
Sequence ::= SequenceIdentifier � old SequenceIdentifier

� SequenceIdentifier ( IdentifierList )
� e � Integer � ( Sequence ) � Sequence ◦ Sequence
� Subrange

Subrange ::= Sequence � Integer : Integer �
Integer ::= IntegerConstant � ElementSelection

� IntegerIdentifier � old IntegerIdentifier
� IntegerIdentifier [ IdentifierList ]
� |Sequence| � ArithmeticCompound

IntegerConstant ::= 〈−〉 Digit+

Digit ::= 0 � 1 � . . . � 9
ElementSelection ::= SequenceIdentifier [ IntegerConstant ]

� (Sequence) [ IntegerConstant ]
ArithmeticCompound ::= Integer Operator Integer

Operator ::= + � − � ∗ � /
IdentifierList ::= Identifier � Identifier , IdentifierList

Identifier ::= IntegerIdentifier � SequenceIdentifier � BooleanIdentifier

Notice that integer functions use square, rather than round, brackets to list
actual arguments. Operator precedence is as follows: ◦, arithmetic operators
(first ∗, /, then +, −), arithmetic comparisons, not, and, or, =⇒, ⇐⇒ . Im-
plication is right-associative and all other binary operators are left-associative.
The old keyword can only appear in postconditions, directly in front of global
variable identifiers (of any type).

3.4 Instructions

qfis includes instructions for assignment, conditional (or nondeterministic)
branch, loop, intermediate assertions (assert, assume), nondeterministic shuf-
fling (havoc), routine call. split is the only non-conventional instruction, de-
signed specifically for sequences: split S into S1, S2, ..., Sn is equivalent to

havoc S1, S2, ..., Sn ; assume S
.
= S1 ◦S2 ◦... ◦Sn

Compound instructions sit on multiple line, or on the same line separated
by a semicolon. Similarly, multiple loop invariant clauses are either separated
by labels or semicolons. A condition (in a conditional or loop) <?> denotes
nondeterministic choice.

Instruction ::= skip
� assert Formula
� assume Formula
� havoc IdentifierList
� split Sequence into SequenceIdentifierList
� IntegerIdentifier := Integer
� SequenceIdentifier := Sequence
� BooleanIdentifier := Formula
� Conditional � Loop � RoutineCall

Conditional ::= if Condition then Compound 〈else Compound 〉 end
Loop ::= until Condition 〈invariant 〈ClauseList〉〉 loop Compound end

Condition ::= Formula � <?>
ClauseList ::= Formula � ClauseList ; Formula � ClauseList Label Formula

RoutineCall ::= call 〈IdentifierList :=〉 RoutineIdentifier 〈( IdentifierList )〉
Compound ::= Instruction � Instruction 〈;〉 Compound

3.5 Routines

Each routine has a signature, with input and output arguments, and optionally
a precondition, a postcondition, a declaration of local variables, a frame. The
input arguments are assumed read-only. Pre and postconditions must mention
only input and arguments variables, or global variables. Global variables that
may be modified by the routine must be listed in its frame (modify).

RoutineDeclaration ::= routine RoutineIdentifier 〈Signature〉 Declarations
Signature ::= 〈( VariableDeclaration )〉 〈: ( VariableDeclaration )〉

Declarations ::= 〈Precondition〉 〈Modify〉 〈Locals〉 Body 〈Postcondition〉 end
Precondition ::= require 〈ClauseList〉

Postcondition ::= ensure 〈ClauseList〉
Modify ::= modify IdentifierList
Locals ::= local VariableDeclaration
Body ::= do 〈Compound〉

A qfis input file consists of declarations of functions, axioms, global variables,
and routines.

Program ::= ProgramElement+

ProgramElement ::= FunctionDeclaration � AxiomDeclaration
� GlobalDeclaration � RoutineDeclaration

GlobalDeclaration ::= global VariableDeclaration

4 VC generation

qfis uses a weakest-precondition calculus defined on primitive instructions as
follows, where φ[x 7→ y] is φ with every occurrence of x replaced by y, and v′ is
a fresh variable of the same type as v.

1 wp(skip, φ) = φ
2 wp(assert F, φ) = F and φ
3 wp(assume F, φ) = F =⇒φ
4 wp(havoc V, φ) = φ[V 7→ V’]
5 wp(X := E, φ) = φ[X 7→ E]

The weakest-preconditions of compound and derived instructions is as fol-
lows.

1 wp(A ; B, φ) = wp(A, wp(B, φ))
2 wp(split S into S1, ..., Sm, φ) =
3 wp(havoc S1, ..., Sm ; assume S

.
= S1 ◦... ◦Sm, φ)

4 wp(if F then T else E end, φ) =
5 (F =⇒wp(T, φ)) and (not F =⇒wp(E, φ))
6 wp(if <?> then T else E end, φ) = wp(T, φ) and wp(E, φ)
7 wp(until F invariant I loop L end, φ) = I

If a loop invariant is not specified, assume I = True.
For routine calls, assume foo has m input arguments A1, . . . , Am, n output

arguments B1, . . . , Bn, precondition Pre (True if not specified), postcondition
Post (True if not specified), and frame G1, . . . , Gh. Then, the weakest precon-
dition for a call to foo uses fresh variables for the output arguments to handle
the cases where some input and output actuals coincide, and saves the value
of global variables before the call in other fresh variables Old localk (reused in
every routine call):

1 wp(call V1, ..., Vn := foo (U1, ..., Um), φ) =
2 wp( assert Pre[A1, ..., Am 7→ U1, ..., Um] ;
3 Old local1 := G1 ; ... ; Old localh := Gh

4 havoc G1, ..., Gh ;
5 assume Post[A1, ..., Am, B1, ..., Bn, old G1, ..., old Gh 7→
6 U1, ..., Um, V’1, ..., V’n, Old local1, ..., Old localh]
7 V1 := V’1 ; ... ; Vn := V’n , φ)

With the weakest precondition, we generate the set VC of verification con-
ditions for foo as follows. Let X1, . . . , Xn be the global variables mentioned in
foo’s postcondition with the old syntax. In the generation of verification con-
ditions, we prepend some assignments to the routine body that store the old
values of X1, . . . , Xn in fresh local variables Old X1, ..., Old Xn:

SaveOld , Old X1 := X1 ; ... ; Old Xn := Xn

Then, a reference old Xk to any such variable in the postcondition becomes a
reference to the alias local variable Old Xk:

PostOld , Post[old X1, ..., old Xn 7→ Old X1, ..., Old Xn]

Correspondingly, we have the following rules.

1 VC(routine foo require Pre do Body ensure Post end) =
2 VC(Body, PostOld) ∪ {Pre =⇒wp(SaveOld ; Body, PostOld)}
3 VC(A ; B, φ) = VC(B, φ) ∪ VC(A, wp(B, φ))
4 VC(until F invariant I loop L end, φ) =
5 {I and not F =⇒wp(L, I)} ∪ {I and F =⇒φ}
6 VC(until <?> invariant I loop L end, φ) =
7 {I =⇒wp(L, I)} ∪ {I =⇒φ}

and, for all other instructions Inst, VC(Inst, φ) = ∅.

5 CVC3 axiomatization

qfis maps integer sequences to Lisp-like lists in CVC3:

DATATYPE IntList = nil | cons(head: INT, tail: IntList) END;

The singleton sequence n for an integer term n becomes cons(n, nil).

5.1 Concatenation

Concatenation is the function cat : (IntList , IntList ) → IntList with the fol-
lowing axioms (Tarski’s editor axioms is needed only in the most complex in-
ference, hence commented out for efficiency by default).

%% Congruence
∀ (h: INT, t, o: IntList ): cat(cons(h, t), o) = cons(h, cat(t, o));

%% TC1
∀ (l : IntList ): cat(nil, l ) = l ;
∀ (l : IntList ): cat(l , nil) = l ;

%% TC2
∀ (x, y, z: IntList ): cat(x, cat(y, z)) = cat(cat(x, y), z);

%% TC3: Tarski’s editor axiom
∀ (x, y, u, v: IntList ): cat(x, y) = cat(u, v) ⇒ ∃ (w: IntList ):

(cat(x, w) = u ∧cat(w, v) = y) ∨ (cat(u, w) = x ∧cat(w, y) = v);
%% TC4−6
∀ (x, y: IntList ): cat(x, y) = x ⇔ y = nil;
∀ (x, y: IntList ): cat(x, y) = y ⇔ x = nil;
∀ (x, y: IntList ): cat(x, y) = nil ⇒ x = nil ∧ y = nil;

5.2 Length

Length has signature length : IntList → NONNEGATIVE INT and the usual
inductive definition (where NONNEGATIVE INT are the nonnegative integers)
and congruence relation with respect to concatenation.

∀ (l : IntList ): length (l ) = 0 ⇔ l = nil;
∀ (l : IntList ): l 6=nil ⇒ length (l ) = 1 + length(tail(l ));
∀ (x, y: IntList ): length (cat(x, y)) = length(x) + length(y);

5.3 Element selection

For every element selection A[p] with p >0 appearing in the qfis formulas, the
translation defines a position function atp: IntList → INT with the axioms:

∀ (x: IntList ): Γ(x, p) ⇒ atp(x) = head(tailp−1(x));
∀ (x, y: IntList ): Γ(x, p) ⇒ at2(cat(x, y)) = at2(x);

%% For each integer a, b such that a + b = p, a >0, b >0
∀ (x, y: IntList ):

Γ(x, a) ∧ taila(x) = nil ∧Γ(y, b) ⇒at2(cat(x, y)) = atb(y)

Γ(x, p) is a shorthand for the formula denoting: “x has at least p elements”:

Γ(x, p) =

{
x 6=nil p = 1

tailp−1(x)6=nil ∧Γ(x, p−1) p > 1

and tailn denotes the function tail applied n times (and tail0(x) is just x).
For every element selection A[p] with p≤0 appearing in the qfis formulas,

the translation defines a position function at bottomq: IntList → INT where
q = −p with the axioms:

∀ (x: IntList ): Γ(x, q+1) ∧ tailq+1(x) = nil ⇒at bottomq(x) = head(x);
∀ (x: IntList ): Γ(x, q+2) ⇒at bottomq(x) = at bottomq(tail(x));

5.4 Subranges

For every subrange A�a, b� with 1≤a <b appearing in the qfis formulas, the
translation defines a function rangea b: IntList → IntList with the axioms:

∀ (x: IntList ): Λ(x, b) ⇒ rangea b(x) = nil;
∀ (x: IntList ): Γ(x, b) ⇒ rangea b(x) = Π(x, a, b);

Λ(x, p) is a shorthand for the formula denoting: “x has less than p elements”:

Λ(x, p) =

{
x = nil p = 1

Λ(x, p−1)∨ tailp−1(x)= nil p > 1

Π(x, p, q) is a term corresponding to the subrange from p to q, constructed over
the list datatype:

Π(x, p, q) =

{
cons(head(tailp−1(x)), nil) p = q

cons(head(tailp−1(x)),Π(p+1, q)) p < q

For every subrange A�a, b� with 0 ≥ a >b appearing in the qfis formu-
las, the translation defines a function range bottomp q: IntList → IntList where
p = −a, q = −b with the axioms:

∀ (x: IntList ): Λ(x, q+1) ⇒range bottomp q(x) = nil;
∀ (x: IntList ): Γ(x, q+1) ∧ tailq+1(x) = nil

⇒ range bottomp q(x) = Π(x, 1, q−p);
∀ (x: IntList ): Γ(x, q+2) ⇒range bottomp q(x) = range bottomp q(tail(x));

For every subrange A�a, 0� with 1≤a appearing in the qfis formulas, the
translation defines a function rangea 0: IntList → IntList with the axioms:

∀ (x: IntList ): Λ(x, b) ⇒ rangea 0(x) = nil;

∀ (x: IntList ): Γ(x, b) ⇒ rangea 0(x) = tailb−1(x);

Finally, a subrange A�a,a� is simply treated as the element selection A[a].

6 A tutorial example

In this tutorial example, we are proving the (partial) correctness of an implemen-
tation of MergeSort working on sequences of integers. The complete example
is available qfis’s distribution, in the subdirectory examples with several other
examples. You are welcome to submit your own examples to the author.

MergeSort takes any integer sequence as input and returns the same sequence
sorted. In this example we only specify that the output is sorted and that it
has the same length as the input. Correspondingly, we introduce the following
declaration in qfis.

routine merge sort (A: SEQUENCE): (Result: SEQUENCE)
do
ensure

sorted: sorted? (Result)
same size: |A| = |Result|

end

We declare the predicate sorted? as a logic function with the right signature:

function sorted? (SEQUENCE): BOOLEAN

We also give sorted? the intended semantics with an inductive definition. Notice
that we should also specify how sorted? works on concatenations: the concate-
nation of two sequences is sorted iff both sequences are sorted and the first
sequence’s last element and the second sequence’s first elements are sorted.

axiom sorted?(e)
axiom forall (i: INTEGER) (sorted?(i))
axiom forall (L: SEQUENCE)

(|L| >1 =⇒(sorted?(L) ⇐⇒ L[1]≤L[2] and sorted?(L�2:0�)))
axiom forall (X, Y: SEQUENCE)

(|X| >0 and |Y| >0 =⇒
(sorted?(X) and sorted?(Y) and sorted?(X[0] ◦Y[1]) ⇐⇒ sorted?(X ◦Y)))

Now, let us add a basic implementation of merge sort. If the input has zero
or one element, it is already sorted. Otherwise, the algorithm works recursively:
it splits the input into two sequences, sorts each sequence, and then merges
the result. Notice that the way the input is split into two subsequences is not
relevant for partial correctness but only for progress (and performance).

routine merge sort (A: SEQUENCE): (Result: SEQUENCE)
local L, R: SEQUENCE
do
if |A| ≤1 then

Result := A
else
split A into L, R
call L := merge sort (L)
call R := merge sort (R)
call Result := merge(L, R)

end
ensure

sorted: sorted? (Result)
same size: |A| = |Result|

end

merge takes two sorted sequences and returns a merged sorted sequence:

routine merge (X, Y: SEQUENCE): (M: SEQUENCE)
require

sorted X: sorted?(X)
sorted Y: sorted?(Y)

do
ensure

sorted: sorted? (M)
same size: |M| = |X| + |Y|

end

Calling qfis on the current program results in a successful verification of
merge sort and a failed verification of merge, whose implementation we haven’t
provided yet: qfis stops at the first postcondition of merge which it cannot
establish.

> qfis merge_sort.qfis

This is QFIS --- a verifier for the theory of quantifier-free integer sequences.

Routine merge_sort verified successfully.

Routine merge not verified:
Could not verify VC # 2 -- annotation on line 63

2 routines, 1 verified, 1 errors.

Let us go on and provide an implementation of merge: at every iteration of
the loop the smallest of the elements in L and R (initialized to X and Y because
input arguments are read-only) is added at the end of M. Finally, if any of the
two sequences is not empty after the loop, it is also added at the end of M.

routine merge (X, Y: SEQUENCE): (M: SEQUENCE)
local

L, R: SEQUENCE
do

L := X ; R := Y ; M := e
until |L| = 0 or |R| = 0
loop

if L[1] >R[1] then

M := M ◦R[1] ; R := R�2:0�
else

M := M ◦L[1] ; L := L�2:0�
end

end
if |L| >0 then M := M ◦L else M := M ◦R end

end

qfis still cannot verify merge because we have provided no loop invariants.
In particular, the first error it encounters:

Routine merge not verified:
Could not verify closing of loop on line 38 -- annotation on line 55

means that it cannot establish that the loop invariant is strong enough to guar-
antee the postcondition.

Let us add a loop invariant: M, L, R are always sorted, and additionally the
first element of both L and R is no smaller than the last element of M. Another
clause states the invariance of the combined length of M, L, and R.

until |L| = 0 or |R| = 0
invariant

sorted M: sorted?(M)
sorted L: sorted?(L)
sorted R: sorted?(R)
L smaller: M[0]≤L[1]
R smaller: M[0]≤R[1]

size: |X| + |Y| = |M| + |L| + |R|
loop

...
end

Verification still fails, now the errors reported are about the loop invariant
clauses L smaller and R smaller for which qfis cannot verify consecution (in-
ductiveness). This is because the clauses must be evaluated before and after a
generic iteration of the loop to verify consecution, but M is empty in the first
iteration (hence M[0] is undefined) and L or R is empty after the last iteration
of the loop (hence L[1] or R[1] is undefined). We weaken the two clauses to:

L smaller: |M| >0 and |L| >0 =⇒M[0]≤L[1]
R smaller: |M| >0 and |R| >0 =⇒M[0]≤R[1]

qfis can verify the latest version correctly:

> qfis merge_sort.qfis

This is QFIS --- a verifier for the theory of quantifier-free integer sequences.

2 routines, 2 verified, 0 errors.

A final consistency check is the smoke test, which tries to see if the axioms
(of sorted?) and the background theories used are consistent (i.e., False cannot
be inferred).

> qfis -s merge_sort.qfis

This is QFIS --- a verifier for the theory of quantifier-free integer sequences.

Smoke test successful: the axioms and the background theories seem consistent.

12



7 Command line options

The help screen of qfis is the following:

This is QFIS --- a verifier for the theory of quantifier-free integer sequences.

Usage: qfis [options] <input_file>
where [options] is one or more of the following.

-h display this notice.
-s perform a smoke test (consistency of the axioms).
-w=<wdir> set <wdir> as the work directory (default: current dir).
-d=<ddir> set <ddir> as the declarations directory

(default: /home/user/tools/qfis/prelude/-- $QFIS_PRELUDE is set).
-t=N give up after N seconds for each verification condition (default: 10).
-v parse and typecheck only (no verification).

Notice that the declarations directory is set as recommended in the installation
instructions. We already showed the smoke test feature in the tutorial example.

Work directory. The work directory is where qfis stores the generated CVC3
input files. The files are not removed after verification, so that you can inspect
them (for lower-level debugging or to see exactly the problem with a failed
verification attempt).

Timeout. The timeout is important for non-trivial verifications, because qfis
treats a timed out run of CVC3 as a failed verification. This is consistent with
the observation that CVC3 is usually rather efficient to determine validity but
it often saturates all the available memory when trying to establish “Invalid”
(or “Unknown”). If you go for complex verification conditions you may have to
give CVC3 some more time to try to prove validity.

